
Logic Programming

 Logic programming is the use of 
 logic as a declarative representation language
 Backward chaining as inference rule

 Logic Programming is the basis of the programming language 
PROLOG

3KE&BI: Logic Programming

Logic Programs – A Sequence of Horn Clauses 

 The sentences of a logic program are Horn clauses
 Facts: H
 Rules: H  B1 ∧ B2 ∧ … ∧ Bn

 A Horn clause without any head H is called a query
 Query:  B1 ∧ B2 ∧ … ∧ Bn

 Queries are not part of a logic program, they start 
the inference

KE&BI: Logic Programming

Knowledge Bases in PROLOG
 PROLOG (= PROgramming in LOGic) is a programming language based on Horn 

clauses

 A Prolog knowledge base consists of clauses. 

 Clauses are either facts or rules. Each clause ends with a period "."

 Facts consist of just one literal:

 Rules consist of a head and a body

 Head and body are separated by the symbol ":-" which means „“

 Literals in the body of a rule are separated by comma „ ,“
(the comma is equivalent to the logical AND or „∧“)

 The second rule can be read as: is_strong_sailor(popeye) is true of 
is_a_sailor(popeye) is true and eats(popeye,spinach) is true.

KE&BI: Logic Programming

is_strong(popeye) :- eats(popeye,spinach).
is_strong_sailor(popeye) :- is_a_sailor(popeye),

eats(popeye,spinach).

is_a_dog(pluto).
is_a_sailor(popeye).
eats(popeye,spinach).

Predicates and Literals

 Predicates are the building blocks of clauses

 Predicates have a name and arguments (parameters). Arity is the 
number of arguments.

 Predicates combine values which “make sense” together (are true)

 Examples:
 person(peter)
 married(peter, cindy)
 appointment(knut, “AB1”, “Lecture KE”)
 not female(knut)

 Literals are predicates and negated predicates

6KE&BI: Logic Programming

Predicate Expressions (Literals)
 Clauses contain literals (predicate expressions), which consist of a predicate symbol and 

arguments. The arguments are called terms, e.g.
eats(popeye,spinach). eats is a predicate symbol, 

popeye and spinach are terms 

 Predicates can have arbitrary arity. Arity is the number of arguments:
it_rains/0 has arity 0.
eats/2 has arity 2.
eats_spinach/1 has arity 1.

 The facts friends(popeye,pluto,garfield) and
friends(pluto,mickey)

define two different predicates, namely friends/3 (arity 3) and friends/2 (arity 2).

 The predicate expressions eats(popeye,spinach) and
eats_spinach(popeye)

express the same statements but are not equivalent

 The order of the arguments is significant: father(john,paul) is not the same as 
father(paul,john). We determine which argument position should stand for what, but 
then we have to keep it:

7KE&BI: Logic Programming

it_rains.
eats(popeye,spinach).
eats_spinach(popeye).



© Prof. Dr. K. Hinkelmann, Prof. Dr. H. Wache

Terms

 The basic data structure in Prolog are terms. They are 
arguments of predicates.

 Terms are either simple or compound.

 Simple terms in Prolog are constants and variables

 The constants are symbols and numbers.

 Compound terms are either complex terms or lists. 

8KE&BI: Logic Programming

Simple Terms: Atoms, Numbers, Variables

Atoms are strings that begin with lowercase letters and consist only of letters, 
numbers, and the underscore, or strings that are enclosed in quotation 
marks:
popeye, dog13XYZ, my_dog, "Lea?! @", 'Homer Simpson'

Numbers are integers or floats:
123, 89.5, 0, -323

Variables are strings that begin with a capital letter or an underscore and consist 
only of letters, numbers, and the underscore:
X, Variable, _x, _123, Hund_123, _

Hints:
 Terms should always be 'speaking'.
 The _ variable, which consists only of the underscore, is the anonymous variable.

9KE&BI: Logic Programming

Variables

 Variables can be used in facts, rules and queries.

 Same variables stand for the same values

 The clause exists(X). is a universal fact (fact with an open variable).

exists(X).

likes(Everybody, mickey)-

is_strong(X) :- has_trained(X).
friend(X,Y) :- friend(Y,X).

 Compound terms consist of a functor and any number of arguments.
 The functor is always an atom.
 The arguments are simple or complex terms.

 Functors – like predicate symbols – have an arity (number of 
arguments).

 Examples of complex terms:
loves(mary,father(mary)) 
friends(X,father(father(popeye)))

 Note: In the second example, father/1 is a function symbol and not 
a predicate symbol

Complex Terms

11

Quelle: Wiebke Petersen, Grundkurs Prolog, HHU  Düsseldorf, https://user.phil.hhu.de/~petersen/WiSe2324_Prolog/WiSe2324_Prolog.html

"Mary loves her father"

Queries about Facts

 Inference in Prolog starts with a query. The system concludes whether the 
statement is true.

 Requests are made to the interpreter in the console and evaluated. 

 A query about facts just checks whether the literal in in the knowledge 
base:

is_a_dog(pluto).
is_a_dog(snoopy).
is_a_sailor(popeye).
eats(popeye,spinach).

?- is_a_dog(pluto).
true

?- eats(popeye, spinach).
true

Queries with Variables

 If a query contains variables, the interpreter tries to instantiate the variable 
(i.e. assign a value) in such a way that the statement becomes true

 The assignment of the variables is displayed as a response

 By entering the semicolon (or clicking "Next") the interpreter looks for 
more answers

13KE&BI: Logic Programming

is_a_sailor(popeye).
eats(popeye,spinach).
likes(pluto,mickey).
likes(mickey,pluto).
likes(minnie,mickey).
likes(mickey,minnie).

?- eats(popeye,X).
X = spinach

?- likes(X,Y).
X = pluto, Y = mickey;
X = mickey, Y = pluto;
X = minnie, Y = mickey;
X = mickey, Y = minnie;
no

Exercises 
 Write as a logic program

 john is a person
 peter and mary are persons
 fhnw is a university 
 john is matriculated at fhnw
 A student is a person who is matriculated at a university.

 Queries
 Is john a student?

KE&BI: Logic Programming

Exercise

 Represent in Prolog
 Peter is the Father of Mary
 Peter is the Father of John
 Mary is the Mother of Mark
 Jane is the Mother of Jane

 Write rules for
 grandfather
 grandmother
 sibling (father or mother in common)

KE&BI: Logic Programming

 All clauses, whose head have the same predicate symbol 
and same arity define a predicate.

Predicates

18

living_being(X)  :- mammal(X).
living_being(X)  :- bird(X).

mammal(X) :- elephant(X). 
mammal(X) :- cat(X).

bird(X)      :- penguin(X).

elephant(clyde).

penguin(tweety).

Exercises 
 Write as a logic program

 knut is a person
 "KEBI" is a course
 courses belong to study programs
 bachelor and master are study programs
 computer science is a master program
 "KEBI" belongs to computer science
 courses are taught by teachers
 john attends course "KEBI"
 students are attending courses
 a student who a attends a course of a master program is a master student
 knut teaches "KEBI"

 Queries
 Is john a master student?
 Is knut a teacher?

KE&BI: Logic Programming

Inference Procedure

21KE&BI: Logic Programming

Reasoning in Prolog

 Prolog's principle of automatic reasoning is based on
 the principle of unification and
 backward chaining with backtracking. 

 To prove a target clause, Prolog tries to unify the clause with 
the facts and rule heads given in the knowledge base.

 If the query contains variables, a valid variable assignment 
(substitution) must be found.

22KE&BI: Logic Programming

Inference Procedure for Logic Programming
Let resolvent be the query ?- Q1, …, Qm

While resolvent is not empty do
1. Choose a query literal Qi from resolvent.
2. Choose a renamed1 clause H :- B1, …, Bn from P such that 

Qi and H unify with an most general unifier σ , i.e. Qiσ = Hσ
(Head Unification)

3. If no such Qi and clause exist, then backtrack
4. Remove Qi from the resolvent 
5. Add B1, …, Bn to the resolvent

6. Add σ to σall

7. Apply substitution σ to the resolvent and go to 1.
If resolvent is empty, return σall, else return failure.

1 Renaming means that the variables in the clause get new unique identifiers

KE&KT: Logic Programming



© Prof. Dr. K. Hinkelmann, Prof. Dr. H. Wache

Unification

24KE&BI: Logic Programming

 A predicate from a query must be unifiable with the head of a clause.

 Query: ?- mortal(socrates).

 Clause:    mortal(X) :- human(X).

 Head Unification
 predicate symbols are equal
 Substitution: X=socrates

 New query: ?- human(socrates).

Head Unification

25

Source: Wiebke Petersen, Grundkurs Prolog, HHU  Düsseldorf, https://user.phil.hhu.de/~petersen/WiSe2324_Prolog/WiSe2324_Prolog.html

Unification

 Two expressions Q and H unify if there exists a substitution σ
for any variables in the expressions so that the expressions 
are made identical (Qσ = Hσ)

Unification Rules

 A constant unifies only with itself

 Two structures unify if and only if 
 they have the same (function or) predicate symbol and the 

same number of arguments, and 
 the corresponding arguments unify recursively

 An unbound variable unifies with anything

KE&BI: Logic Programming

Substitution

 A substitution is a finite set of the form σ = {v1/ t1, . . . , vn / tn}
 vi ’s: distinct variables.
 ti ’s: terms with ti ≠ vi .

 Applying a substitution σ to an expression E means to 
replace each occurence of a variables vi with the value ti

 Example: E = p(X, Y, f (a))
σ = {X / b, Y / Z}
Eσ = p(b, Z, f (a))

E = father(peter,X)
σ = {X / mary} 
Eσ = father(peter,mary)

KE&BI: Logic Programming

Unifier

 A substitution σ is a unifier of expressions E and F iff 

  Eσ= Fσ

 Example: Let E and F be two expressions:
 E = f (x, b, g(z)),
 F = f (f (y), y, g(u)).

 Then  σ = {x / f (b), y / b, z / u} is a unifier of E and F:
 Eσ = f (f (b), b, g(u)),
 Fσ = f (f (b), b, g(u)

 A unifier σ of E and F is most general iff  is more general than any other 
unifier of E and F, i.e. for any other unifier ρ there exists a unifier τ such 
that ρ = τ ° σ 

KE&BI: Logic Programming

 In unification, two terms are compared with each other or 
checked whether they can be equated (unified) by a 
suitable variable assignment.

 Unification is a part of reasoning. However, there is also 
the built-in predicate =, which equates two terms.

Unification

29

Quelle: Wiebke Petersen, Grundkurs Prolog, HHU  Düsseldorf, https://user.phil.hhu.de/~petersen/WiSe2324_Prolog/WiSe2324_Prolog.html

Unification Rule
Two terms are unifiable if and only if 
• they are equal, or 
• there is a substitution that assigns values to the 

variables in such a way that the two terms 
become equal

 If one of the terms is a variable, then the variable can be 
substituted with the other term

 If a variable occurs more than once in a term, the variable 
assignment must be compatible everywhere

Unification of Terms

30

SOurce: Wiebke Petersen, Grundkurs Prolog, HHU  Düsseldorf, https://user.phil.hhu.de/~petersen/WiSe2324_Prolog/WiSe2324_Prolog.html

?- X=Y, X=popeye.
X = popeye,
Y = popeye.

?- X=popeye, X=pluto.
false.

?- friend(popeye) = X.
X = friend(popeye).

 Complex terms match exactly when:
1) the terms have the same functor and the same arity, and
2) match all corresponding arguments match, and
3) the variable assignments are compatible with each other.

Unification of Complex Terms

31

Quelle: Wiebke Petersen, Grundkurs Prolog, HHU  Düsseldorf, https://user.phil.hhu.de/~petersen/WiSe2324_Prolog/WiSe2324_Prolog.html

?- food(bread,X) = food(Y,sausage).
X = sausage,
Y = bread.

?- meal(food(bread), drink(beer)) = meal(X,drink(Y)).
X = food(bread),
Y = beer.

?- food(bread,X,beer) = food(Y,sausage,X)
false.

Backward Chaining

32KE&BI: Logic Programming

Rules and Inferences

 If the rule body is true (i.e. can be derived from the 
knowledge base), then the rule head is also true.

 This principle of deduction is called Modus Ponens:

 From the rule is_strong(popeye) :- eats(popeye,spinach). 
and the fact eats(popeye,spinach). the Prolog interpreter 
infers that is_strong(popeye). applies.

33KE&BI: Logic Programming

is_strong(popeye) :- eats(popeye,spinach).
eats(popeye,spinach).

is_strong(popeye).

Rules and Queries

 To answer queries, the rules are applied backwards. 
 If the query matches a fact, the query is true
 If the query matches the header of a rule, the body  becomes 

the new query

34KE&BI: Logic Programming

is_a_sailor(popeye).
eats(popeye,spinach).

is_strong(popeye) :- eats(popeye,spinach).

has_muscles(popeye) :- has_trained(popeye).
has_muscles(popeye) :- is_strong(popeye).

eats(popeye,spinach).
?- eats(popeye,spinach).

?- true.

is_strong(popeye) :- eats(popeye,spinach).
?- is_strong(popeye).

?- eats(popeye,spinach).

Illustrating Backward Chaining

Source: Kerber (2004), http://www.cs.bham.ac.uk/~mmk/Teaching/AI/l2.html

KE&BI: Logic Programming

There are two choices in the inference procedure where a 
decision needs to be made
 Step 1: Selecting the Literal Qi from the Resolvents 
 Solution in Logical Programming: left-most goal 

 Step 2: Choosing a clause
 Solution in logical programming: top-most clause
 The clauses are selected in the order in which they appear.
 Backtracking: If a selected clause does not succeed and there are 

alternative clauses, the next one is selected.

Two Choices in the Inference Procedure

36

 Rule bodies and queries are proven from left to right.

 Only when a proof of the literal i in a rule is found, ther
literal i+1 can be proven

 Example Query:

 Example Rule

 First female(X), is proven and then  sibling(X,Y)

Within a Rule: From left to right

37

Source: Wiebke Petersen, Grundkurs Prolog, HHU  Düsseldorf, https://user.phil.hhu.de/~petersen/WiSe2324_Prolog/WiSe2324_Prolog.html

?- female(X), sibling(X,Y).

sister(X,Y):- female(X), sibling(X,Y).

 Head Unification is performed top down.

 The interpreter searches the database from top to bottom 
to find suitable clauses for proof

 What is the first answer the the query:

Choosing a Clause: Top-Down

38

Quelle: Wiebke Petersen, Grundkurs Prolog, HHU  Düsseldorf, https://user.phil.hhu.de/~petersen/WiSe2324_Prolog/WiSe2324_Prolog.html

eat_spinach(popeye).
has_trained(garfield).
is_strong(X) :- has_trained(X).
is_strong(X) :- eat_spinach(X).

?- is_strong(X).
X = garfield

Search Tree for Depth-Firast Search

39

Source: Wiebke Petersen, Grundkurs Prolog, HHU  Düsseldorf, https://user.phil.hhu.de/~petersen/WiSe2324_Prolog/WiSe2324_Prolog.html

?- dog(pluto).

?- is_mammal(pluto), barks(pluto).

?- has_four_legs(pluto), barks(pluto).

?- barks(pluto).



X = pluto

is_honest(pluto).
has_four_legs(pluto).
barks(pluto).

dog(X) :-
is_mammal(X),
barks(X).

is_mammal(X) :-
has_four_legs(X).

?- dog(pluto).

Search Tree for Depth-Firast Search

40

Source: Wiebke Petersen, Grundkurs Prolog, HHU  Düsseldorf, https://user.phil.hhu.de/~petersen/WiSe2324_Prolog/WiSe2324_Prolog.html

is_honest(pluto).
has_four_legs(pluto).
barks(pluto).

dog(X) :-
is_mammal(X),
barks(X).

is_mammal(X) :-
has_four_legs(X).

?- dog(pluto).

[trace] 8 ?- trace, dog(pluto).
Call: (7) dog(pluto) ?
Call: (8) is_mammal(pluto) ?
Call: (9) has_four_legs(pluto) ?
Exit: (9) has_four_legs(pluto) ?
Exit: (8) is_mammal(pluto) ?
Call: (8) barks(pluto) ?
Exit: (8) barks(pluto) ?
Exit: (7) dog(pluto) ?
true.

A Logic Program and Queries

41KE&BI: Logic Programming

grandfather(X,Z) :- father(X,Y), father(Y,Z).
grandfather(X,Z) :- father(X,Y), mother(Y,Z).

grandmother(X,Z) :- mother(X,Y), father(Y,Z).
grandmother(X,Z) :- mother(X,Y), mother(Y,Z).

sibling(Y,Z) :- father(X,Y), father(X,Z).
sibling(Y,Z) :- mother(X,Y), mother(X,Z).

father(peter,mary).
father(peter,john).

mother(mary,mark).
mother(jane,mary).

Queries : ?- father(peter,john).
?- father(peter,X).
?- grandfather(peter,mark).
?- grandfather(peter,mary).
?- grandfather(peter,S).
?- sibling(X,Y).

Adding Goal to Resolvent

 In step 5 of the Inference procedure the literals of the clause 
are added to the resolvent.

 Depending on whether the literals are added at the beginning 
or the end of the resolvent, we get two different strategies:
 Adding the literals to the beginning of the resolvent gives 

depth-first search.
 Adding the literals to the end of the resolvent gives breadth-

first search.

KE&BI: Logic Programming

Backtracking

43KE&BI: Logic Programming



© Prof. Dr. K. Hinkelmann, Prof. Dr. H. Wache

 Backtracking can be triggered by two causes:
 There is no further clause for the current query predicate.
 An alternative solution is to be calculated.

 In any case, the interpreter goes back to the last branch in 
the proof tree, where alternatives were still open (depth-
first).

Backtracking: Depth-First Search

44

Source: Wiebke Petersen, Grundkurs Prolog, HHU  Düsseldorf, https://user.phil.hhu.de/~petersen/WiSe2324_Prolog/WiSe2324_Prolog.html

?- is_strong(popeye).
?- is_strong(X).

eat_spinach(popeye).
has_trained(garfield).
is_strong(X) :- has_trained(X).
is_strong(X) :- eat_spinach(X).

Search Tree: Decision Point and Backtracking

45

?- is_strong(popeye).

X = popeye

?- has_trained(popeye).

Decision Point

Backtracking

?- eat_spinach(popeye).



X = popeye

?- is_strong(popeye).

eat_spinach(popeye).
has_trained(garfield).
is_strong(X) :- has_trained(X).
is_strong(X) :- eat_spinach(X).

?- is_strong(X).

X = X1

?- has_trained(X1).

Decision Point

Backtracking for 
next

?- eat_spinach(X2).



X1 = garfield



X = X2

X2 = popeye

Backtracking 
for next solution 

fail

?- is_strong(X).

eat_spinach(popeye).
has_trained(garfield).
is_strong(X) :- has_trained(X).
is_strong(X) :- eat_spinach(X).

Backtracking

 Record any decision 
(choose) and its alternative

 If backtracking, then go back 
to the last decision and try 
another option

 When backtracking then roll 
back to the former situation 
(esp. for resolvent and σall)

47KE&KT: Logic Programming

Search Tree: Decision Points

48

Source: Wiebke Petersen, Grundkurs Prolog, HHU  Düsseldorf, https://user.phil.hhu.de/~petersen/WiSe2324_Prolog/WiSe2324_Prolog.html

Prolog‘s Solution: Summary

 Choice of a query literal: 
 leftmost literal first

 Choice of a clause
 Topmost clause first - with backtracking

 Adding new goal to the resolvent
 At the beginning.

KE&BI: Logic Programming

Recursion

50KE&BI: Logic Programming

 Given these facts:

Task

51

costs_less(lolli,icecream).
costs_less(icecream,burger).
costs_less(burger,steak).
costs_less(steak,sushi).

Write rules for cheaper/2.  such that cheaper(X,Y) 
is true, if X costs less than Y.

costs_less:
cheaper:

sushi

steak

burger

icecream

lolli

Nicht-rekursive Definition

52

costs_less:
cheaper:

sushi

steak

burger

icecream

lolli
costs_less(lolli,icecream).
costs_less(icecream,burger).
costs_less(burger,steak).
costs_less(steak,sushi).

cheaper(X,Y):- costs_less(X,Y).

cheaper(X,Y):- costs_less(X,A),
costs_less(A,Y).

cheaper(X,Y):- costs_less(X,A),
costs_less(A,B),
costs_less(B,Y).

cheaper(X,Y):- costs_less(X,A),
costs_less(A,B),
costs_less(B,C),
costs_less(C,Y).

Solution: Recursive Predicate

53

costs_less(lolli,icecream).
costs_less(icecream,burger).
costs_less(burger,steak).
costs_less(steak,sushi).

cheaper(X,Y):- costs_less(X,Y).
cheaper(X,Y):- costs_less(X,Z),

cheaper(Z,Y).

costs_less:
cheaper:

sushi

steak

burger

icecream

lolli

Recursion

 In the knowledge base we see that pluto likes mickey and als mickey likes 
pluto. The same for minnie and mickey.

 Assume that likes is a inverse predicate. How can we avoid to write all 
facts. Assume want to write only one likes fact for a couple and get the 
inverse by inference.

54KE&BI: Logic Programming

likes(pluto,mickey).
likes(mickey,pluto).
likes(minnie,mickey).
likes(mickey,minnie).

likes(pluto,mickey).
likes(minnie,mickey).
likes(X,Y) :- likes(Y,X).

 A predicate is defined recursively when the predicate in the rule head 
is called in one of the defining rules.

 The basic idea is to reduce a common task to a simpler task of the 
same class (loops).

 Recursion makes it possible to write compact predicate definitions 
and avoid redundancy.

Recursive Predicates

55

costs_less(lolli,icecream).
costs_less(ice_cream, burger).
costs_less(burger,steak).
costs_less(steak,sushi).

cheaper(X,Y):- costs_less(X,Y).
cheaper(X,Y):- costs_less(X,Z),

cheaper(Z,Y).

likes(pluto,mickey).
likes(minnie,mickey).
likes(X,Y) :- likes(Y,X).

Declarative Interpretation
– Declarative meaning is the meaning that is 'meant' or 'expressed' when reading the 

knowledge base as a set of logical statements.
– The declarative meaning of a prolog program can be defined as the set of all 

statements that can be logically derived from the knowledge base

Declarative und procedural Interpretation of a 
Knowledge Base

56

Source: Wiebke Petersen, Grundkurs Prolog, HHU  Düsseldorf, https://user.phil.hhu.de/~petersen/WiSe2324_Prolog/WiSe2324_Prolog.html

Procedural Interpretation
– Procedural interpretation is the meaning that comes from what Prolog 'does' with a 

knowledge base.
– The procedural meaning of a Prolog program can be defined as the set of all queries 

(statements) for which the Prolog interpreter finds a variable assignment that results in 
the output true



© Prof. Dr. K. Hinkelmann, Prof. Dr. H. Wache

 First rule: To prove that X is cheaper than Y, it is enough to prove that 
X costs less than Y.

 Second rule: To prove that X is cheaper than Y, this problem can be 
broken down into two sub-problems. We are looking for a Z so that X 
costs less than Z (subproblem 1) and that Z is cheaper than Y 
(subproblem 2).

Procedural Interpretation of Recursive Predicates

57
Quelle: Wiebke Petersen, Grundkurs Prolog, HHU  Düsseldorf, https://user.phil.hhu.de/~petersen/WiSe2324_Prolog/WiSe2324_Prolog.html

cheaper(X,Y):- costs_less(X,Y).
cheaper(X,Y):- costs_less(X,Z),

cheaper(Z,Y).

 As a reminder, Prolog works its way
 through the knowledge base from top to bottom,
within the clauses from left to right.

 How does the order affect the procedural behavior of the 
predicate?

Procedural und declarative Interpretation

58

Quelle: Wiebke Petersen, Grundkurs Prolog, HHU  Düsseldorf, https://user.phil.hhu.de/~petersen/WiSe2324_Prolog/WiSe2324_Prolog.html

parent(john,peter).
parent(mary,john).
parent(susan,mary).

ancestor3(X,Y):- parent(X,Y).
ancestor3(X,Z):- ancestor3(Y,Z),

parent(X,Y).

ancestor1(X,Y):- parent(X,Y).
ancestor1(X,Z):- parent(X,Y),

ancestor1(Y,Z).

ancestor2(X,Z):- parent(X,Y),
ancestor2(Y,Z).

ancestor2(X,Y):- parent(X,Y).

ancestor4(X,Z):- ancestor4(Y,Z),
parent(X,Y).

ancestor4(X,Y):- parent(X,Y).

 Recursive predicates always require at least two clauses: 
 a recursive clause 
 an anchor or exit clause.

 The anchor clause should always precede the recursive clause 
(otherwise there is a risk of an infinite loop).

 In the rule body of the recursive clause, it often makes sense to put 
the recursive call at the end.

Defining "harmless" recursive Predicates

59

Quelle: Wiebke Petersen, Grundkurs Prolog, HHU  Düsseldorf, https://user.phil.hhu.de/~petersen/WiSe2324_Prolog/WiSe2324_Prolog.html

anchor clause
cheaper(X,Y):- 
      costs_less(X,Y).
cheaper(X,Y):- 
      costs_less(X,Z),

cheaper(Z,Y).

Multiple Answers to a Query

 The inference procedure of Prolog computes one solution.

 The user can force the system to compute the next solution by typing a „;“ 
(typing „;“ is interpreted by the system as a fail and thus backtracking is 
started to compute an alternative solution)

 Example:

sibling(Y,Z) :- father(X,Y), father(X,Z).
sibling(Y,Z) :- mother(X,Y), mother(X,Z).

father(peter,mary).
father(peter,john).
father(peter,paul).

?- sibling(X,Y).
X=mary, Y=mary;
X=mary, Y=john;
X=mary, Y=paul;
X=john, Y=mary

KE&BI: Logic Programming

Negation and Cut

61KE&BI: Logic Programming

Negation as Failure

 Prolog allows a form of negation that is called negation as 
failure

 A negated query 

  not Q

 is considered proved if the system fails to prove Q

 Thus, the clause

  alive(X) :- not dead(X)

 can be read as „Everyone is alive if not provably dead“

KE&BI: Logic Programming

The Cut Operator

 Under procedural reading, a logic program consists of a set 
of procedures

 Each procedure consists of a sequence of alternatives

 The inference procedure of Prolog computes all possible 
alternatives for a query

 The cut operator (written as „!“) prevents backtracking. It is a 
special literal that is always true but that stops all other 
alternatives from being applied.

sibling(Y,Z) :- father(X,Y), !, father(X,Z).
sibling(Y,Z) :- mother(X,Y), mother(X,Z).

KE&BI: Logic Programming

 The query ?-risk(X) gives X=high

 When asking for another solution is also give X=low, which is 
wrong.

 How this be avoided?

64

Application of the Cut

return(4).

risk(high) :- return(X), X < 5.
risk(low).

risk2(high) :- return(X), X < 5.
risk2(low) :- return(X), X >= 5.

risk3(high) :- return(X), X < 5, !.
risk3(low).

Example with Cut

65

p(X) :- a(X).

p(X) :- b(X),
c(X),!,
d(X),
e(X).

p(X) :- f(X).

a(1).
b(1).
b(2).
c(1).
c(2).
d(2).
e(2).
f(3).

?-p(X).
X=1;
false.

Defining Negation as Failure with the Cut Operator

 The cut operator can be used to define negation as failure

 If ?- Q can be proved then the query not(Q) fails.

 If Q cannot be proved, the second clause is applied which 
always succeeds.

 If Q can be proved the second clause must not be applied. 
This is assured by the cut: If Q can be proved, then the cut 
prevents backtracking.

not(Q) :- Q, !, fail.

not(Q).

KE&BI: Logic Programming

 Reasons to use the Cut
 Efficiency: Cropping the search space
 Shorter Programs
 Enforcing determinism.
 Modeling of defaults.

 Downsides of the cut
 The cut destroys the declarativity of prolog programs.
 The interpretation of a predicate definition with cuts is usually 

only possible if the order of the proof steps is taken into 
account.

67

Using the Cut

Arithmetics

68KE&BI: Logic Programming

Built-in  Arithmetic: The Operator is/2

 In Prolog there is a set of built-in functions for arithmetics. To apply these function 
there exists a special predicate „is“:

  X is Y  is true when X is equal to the value of Y.

 Built-in functions include: +, –, *, /, //, mod, (// performs integer division)
 Using these functions we can compute a value for terms involving numbers.

 Example:
 ?- X is 7+1.

 Will give the answer X = 8

 The is Predicate works as follows:
 First evaluate the right-hand argument  (after the „is“)
 The result is is unified with the left-hand argument.
 The values of all the variables on the right-hand side of is must be known for 

evaluation to succeed.

KE&BI: Logic Programming

How to use is/2

 The operator forces the second argument to be evaluated 
immediately. Therefore, the second argument must be an 
evaluable arithmetic expression

 If the second argument cannot be evaluated, Prolog aborts 
with an error message

70KE&BI: Logic Programming

?- 3+5 is X.
ERROR: is/2: Arguments are not sufficiently instantiated
?- X is 4+Y.
ERROR: is/2: Arguments are not sufficiently instantiated
?- X is a.
ERROR: Arithmetic: ‘a’ is not a function

Comparison

The comparison operators < (smaller), =<, (less than or equal to), 
> (greater), >= (greater than or equal to), =:= (equal), and =\= 
(unequal) force the immediate evaluation of both arguments

71KE&BI: Logic Programming

?- 1+4 < 3*5.
true.

?- 1+7 =< 3*2.
false.

?- 1+3 =:= 2*2.
true.

?- 1+3 =\= 2*3.
true.

?- X < 3.
ERROR: </2: Arguments are not sufficiently instantiated

Comparison

Equality:

Other Comparisons:

KE&BI: Logic Programming

Lists

73KE&BI: Logic Programming



© Prof. Dr. K. Hinkelmann, Prof. Dr. H. Wache

 Lists are very powerful data structures in Prolog.
 Lists are finite sequences of elements
 Lists can contain different types of terms
 Lists can be nested (lists can have lists as items)
 Difference to sets:
 The order of the elements is important [a,b,c] 6= [b,a,c] 
 The same item can appear multiple times in a list

Lists in Prolog

74

Quelle: Wiebke Petersen, Grundkurs Prolog, HHU  Düsseldorf, https://user.phil.hhu.de/~petersen/WiSe2324_Prolog/WiSe2324_Prolog.html

[mia, vincent, jules, mia]
[mia, 2, mother(jules), X, 1.7]
[]
[mia, [[3,4,paul], mother(jules)], X]

 Two lists are unifiable,
 if they are of the same length, and
 if the corresponding list items are unifiable.

 The length of a list is the number of items it contains

Unification of Lists

75

Quelle: Wiebke Petersen, Grundkurs Prolog, HHU  Düsseldorf, https://user.phil.hhu.de/~petersen/WiSe2324_Prolog/WiSe2324_Prolog.html

?- [a,b,X]=[Y,b,3].
X = 3, Y = a

?- [[a,b,c],b,3]=[Y,b,3].
Y = [a, b, c]

?- [a,b,c] = X. 
X=[a,b,c]

?- [a,b,X,c]=[Y,b,3].
false.

?- [a,c,3]=[Y,b,3].
false.

Listenzerlegung in Prolog

 The list constructor ‘|’ divides a list in head and tail.
 The head is the first element of the list
 The tail is the rest of the list. It is itself a list

 A empty list has no head and therfore cannot be split

 The '|' can also seperate more than one leading elements

76

?- [Head|Tail] = [mia, vincent, jules, mia].
Head = mia,
Tail = [vincent, jules, mia].

?- [Head|Tail] = [].
false.

?- [First,Second|Tail] = [mia, vincent, jules, mia].
First = mia,
Second = vincent,
Tail = [jules, mia].

 The variable ‘_’ is the anonymous variable in Prolog.

 It is always used when a value is no longer needed later.

 Unlike other variables, each occurrence of the anonymous 
variable is independent of the others. So it can be 
initialized differently again and again:

Anonymous Variable

77

Quelle: Wiebke Petersen, Grundkurs Prolog, HHU  Düsseldorf, https://user.phil.hhu.de/~petersen/WiSe2324_Prolog/WiSe2324_Prolog.html

?- [_,X2,_,X4|_] = [mia, vincent, jules, mia, otto, lena].
X2 = vincent,
X4 = mia.

 Define a predicate member/2, which tests whether an element 
occurs in a list

Task

78

Note: member/2 is a predefined predicate in some prolog 
systems that is loaded automatically. To define our own 
predicate, we use our own name, e.g. my_member/2.

 member/2 is a recursively defined predicate that checks whether an item 
appears in a list:

 The fact member(X,[X|_]). says that something is an element of a list if it is 
the first item (the head) of the list.

 The rule member(X,[_|T]):- member(X,T). says that something is an 
element of a list if it is an element of the remainder list (the tail).

 Each item in a list is either the first item or an item in the tai

Predicate member/2

79

Quelle: Wiebke Petersen, Grundkurs Prolog, HHU  Düsseldorf, https://user.phil.hhu.de/~petersen/WiSe2324_Prolog/WiSe2324_Prolog.html

% member/2, member(Term,List)
member(X,[X|_]).
member(X,[_|T]) :- member(X,T).

 member/2 can be used to

 test whether an element occour in a list (the first 
argument is a constant) 

 find the elements of a list (the first argument is a variable)

Declarative Application of member/2 

80
Quelle: Wiebke Petersen, Grundkurs Prolog, HHU  Düsseldorf, https://user.phil.hhu.de/~petersen/WiSe2324_Prolog/WiSe2324_Prolog.html

?- member(1,[1,2,3]).
?- member(1,L).

?- member(X,[1,2,3]).

Four predicates for recursive lists processing:

member/2 access to list elements.
member(Element,List)

append/3 concatenation of lists.
append(List1,List2,Konkatlist)

delete/3 deleting list elements or adding elements to a list.
delete(Element,List,ListDeleted)

reverse/2 reverting a list
reverse(List,ListReversed)

Basic Predicates for List Manipulation

81

The predicate append/3 ca

 test, whether a list is the concatenation of two lists:

 concatenate two lists :

 divide lists :

 Petersen

Concatenating Lists: append/3

82

append([],L,L).
append([H|T1],L2,[H|T3]) :- append(T1,L2,T3).

?- append([1,2,3],[4,5,6],[1,2,3,4,5,6]).
true.

?- append([1,2,3],[4,5,6],L).
L = [1,2,3,4,5,6].

?- append(L,[4,5,6],[1,2,3,4,5,6]).
L = [1,2,3].

?- append([1,2,3],L,[1,2,3,4,5,6]).
L = [4,5,6].

?- append(X,Y,[a,b,c]).
X = [], Y = [a,b,c];
X = [a], Y = [b,c];
X = [a,b], Y = [c];
X = [a,b,c], Y = [];

The predicate append/3 can be used to determine sublists:

 Prefix of a list [a,b,c,d]: [],[a],[a,b],[a,b,c],[a,b,c,d]

 Suffix os a list [a,b,c,d]: [],[d],[c,d],[b,c,d],[a,b,c,d]

 Sublists of a list [a,b,c]: [],[a],[a,b],[a,b,c],[b],[b,c],[c]

Suffix, Prefix and Sublist

83

prefix(P,L) :- append(P,_,L).

suffix(S,L) :- append(_,S,L).

sublist(SL, L) :- prefix(P,L), suffix(SL,P).

 delete/3 relates a term and two lists sich that list2 is the 
result of deleting one occurence of the term in list 1

Delete a list element: delete/3

84

% delete/3,     delete(Term,List1,List2)

delete(X,[X|T],T).
delete(X,[H|T1],[H|T2]):-

delete(X,T1,T2).

?- delete(b,[a,b,c],[a,c]).
true.

?- delete(c,[a,b,c],X).
X=[a,b]

?- delete(X,[a,b,c,d],[a,b,d]).
X = c

?- delete(1,X,[a,b,c]).
X = [1, a, b, c] ;
X = [a, 1, b, c] ;
X = [a, b, 1, c] ;
X = [a, b, c, 1] . 

Length of a List

85

?- len1([a,[b,e,[f,g]],food(cheese),X],4).

true.

?- len1([a,b,a],X).

X=3.

% len1/2 , len1(List, Length)

length([],0).
length([_|T],N):- length(T,X),

N is X+1.

